// File: contracts/FundingExecutor.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.11;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/utils/Strings.sol";
import "./interfaces/GeneralTokenVesting.sol";
import "./interfaces/Finance.sol";

/**
 * @title FundingExecutor
 * @dev allow a whitelisted set of addresses to Fund SARCO, via stablecoins (USDC),
 * @dev exchange rate set w/ each traunche/group
 */
contract FundingExecutor is Ownable {
    using SafeERC20 for IERC20;

    uint256 private constant USDC_TO_SARCO_PRECISION = 10**18;
    uint256 private constant SARCO_TO_USDC_DECIMAL_FIX = 10**(18 - 6);
    uint256 public immutable sarcoAllocationsTotal;
    uint256 public immutable offerExpirationDelay;
    uint256 public immutable vestingEndDelay;
    IERC20 public immutable usdcToken;
    IERC20 public immutable sarcoToken;
    GeneralTokenVesting public immutable generalTokenVesting;
    Finance public immutable sarcoDao;

    uint256 public offerStartedAt;
    uint256 public offerExpiresAt;

    uint256[] public usdcToSarcoRates;
    struct FunderInfo {
        uint256 sarcoAllocation;
        uint256 usdcToSarcoRateIndex;
    }

    mapping(address => FunderInfo) public funders;

    // The purchase has been executed exchanging USDC to vested SARCO
    event FundingExecuted(
        // the address that has received the vested SARCO tokens
        address indexed sarcoReceiver,
        // the number of SARCO tokens vested to sarcoReceiver
        uint256 sarcoAllocation,
        // the amount of USDC that was invested and forwarded to the DAO
        uint256 usdcCost
    );

    // Creates a window of time which the whitelisted set of addresses may invest in SARCO
    event OfferStarted(
        // Window start time
        uint256 startedAt,
        // Window end time
        uint256 expiresAt
    );

    // If tokens have not been invested after time window, the DAO can recover tokens
    event TokensRecovered(
        // Amount of Tokens
        uint256 amount
    );

    /**
     * @dev inits/sets sarco funding enviorment
     * @param _usdcToSarcoRates Array of usdc/sarco rates
     * @param _vestingEndDelay Delay from the purchase moment to the vesting end moment, in seconds
     * @param _offerExpirationDelay Delay from the contract deployment to offer expiration, in seconds
     * @param _sarcoFunders  List of valid SARCO investor
     * @param _sarcoAllocations List of SARCO token allocations, should include decimals 10 ** 18
     * @param _sarcoRateIndex group which Sarco investor belongs to
     * @param _sarcoAllocationsTotal Checksum of SARCO token allocations, should include decimals 10 ** 18
     * @param _usdcToken USDC token address
     * @param _sarcoToken Sarco token address
     * @param _generalTokenVesting GeneralTokenVesting contract address
     * @param _sarcoDao Sarco DAO contract address
     */
    constructor(
        uint256[] memory _usdcToSarcoRates,
        uint256 _vestingEndDelay,
        uint256 _offerExpirationDelay,
        address[] memory _sarcoFunders,
        uint256[] memory _sarcoAllocations,
        uint256[] memory _sarcoRateIndex,
        uint256 _sarcoAllocationsTotal,
        address _usdcToken,
        address _sarcoToken,
        address _generalTokenVesting,
        address _sarcoDao
    ) {
        require(
            _usdcToSarcoRates.length > 0,
            "FundingExecutor: array must be greater than 0"
        );
        require(
            _vestingEndDelay > 0,
            "FundingExecutor: endDelay must be greater than 0"
        );
        require(
            _offerExpirationDelay > 0,
            "FundingExecutor: offerExpiration must be greater than 0"
        );
        require(
            _sarcoFunders.length == _sarcoAllocations.length &&
                _sarcoAllocations.length == _sarcoRateIndex.length,
            "FundingExecutor: purchasers, allocations, indexes lengths must be equal"
        );
        require(
            _usdcToken != address(0),
            "FundingExecutor: _usdcToken cannot be 0 address"
        );
        require(
            _sarcoToken != address(0),
            "FundingExecutor: _sarcoToken cannot be 0 address"
        );
        require(
            _generalTokenVesting != address(0),
            "FundingExecutor: _generalTokenVesting cannot be 0 address"
        );
        require(
            _sarcoDao != address(0),
            "FundingExecutor: _sarcoDao cannot be 0 address"
        );

        // Set global variables
        usdcToSarcoRates = _usdcToSarcoRates;
        vestingEndDelay = _vestingEndDelay;
        offerExpirationDelay = _offerExpirationDelay;
        sarcoAllocationsTotal = _sarcoAllocationsTotal;
        usdcToken = IERC20(_usdcToken);
        sarcoToken = IERC20(_sarcoToken);
        generalTokenVesting = GeneralTokenVesting(_generalTokenVesting);
        sarcoDao = Finance(_sarcoDao);
        uint256 allocationsSum = 0;
        for (uint256 i = 0; i < _sarcoFunders.length; i++) {
            require(
                _sarcoFunders[i] != address(0),
                "FundingExecutor: Funder cannot be the ZERO address"
            );
            require(
                funders[_sarcoFunders[i]].sarcoAllocation == 0,
                "FundingExecutor: Allocation has already been set"
            );
            require(
                _sarcoAllocations[i] > 0,
                "FundingExecutor: No allocated Sarco tokens for address"
            );
            require(
                _usdcToSarcoRates[_sarcoRateIndex[i]] > 0,
                "FundingExecutor: _usdcToSarcoRates must be greater than 0"
            );
            funders[_sarcoFunders[i]] = FunderInfo(
                _sarcoAllocations[i],
                _sarcoRateIndex[i]
            );
            allocationsSum += _sarcoAllocations[i];
        }
        require(
            allocationsSum == _sarcoAllocationsTotal,
            "FundingExecutor: AllocationsTotal does not equal the sum of passed allocations"
        );

        // Approve SarcoDao - PurchaseExecutor's total USDC tokens (Execute Purchase)
        IERC20(_usdcToken).approve(_sarcoDao, type(uint256).max);

        // Approve full SARCO amount to GeneralTokenVesting contract
        IERC20(_sarcoToken).approve(
            _generalTokenVesting,
            _sarcoAllocationsTotal
        );

        // Approve SarcoDao - Funding Executor's total SARCO tokens (Recover Tokens)
        IERC20(_sarcoToken).approve(_sarcoDao, _sarcoAllocationsTotal);
    }

    function _getUsdcCost(uint256 sarcoAmount, uint256 index)
        internal
        view
        returns (uint256)
    {
        return
            ((sarcoAmount * USDC_TO_SARCO_PRECISION) /
                usdcToSarcoRates[index]) / SARCO_TO_USDC_DECIMAL_FIX;
    }

    function offerStarted() public view returns (bool) {
        return offerStartedAt != 0;
    }

    function offerExpired() public view returns (bool) {
        return block.timestamp >= offerExpiresAt;
    }

    /**
     * @notice Starts the offer if it 1) hasn't been started yet and 2) has received funding in full.
     */
    function _startUnlessStarted() internal {
        require(
            offerStartedAt == 0,
            "FundingExecutor: Offer has already started"
        );
        require(
            sarcoToken.balanceOf(address(this)) == sarcoAllocationsTotal,
            "FundingExecutor: Insufficient Sarco contract balance to start offer"
        );

        offerStartedAt = block.timestamp;
        offerExpiresAt = block.timestamp + offerExpirationDelay;
        emit OfferStarted(offerStartedAt, offerExpiresAt);
    }

    function start() public {
        _startUnlessStarted();
    }

    /**
     * @dev Returns the Sarco allocation and the USDC cost to invest the Sarco Allocation of the whitelisted Sarco investor
     * @param sarcoReceiver Whitelisted Sarco Investor
     * @return A tuple: the first element is the amount of SARCO available for purchase (zero if
        the purchase was already executed for that address), the second element is the
        USDC cost of the purchase.
     */
    function getAllocation(address sarcoReceiver)
        public
        view
        returns (uint256, uint256)
    {
        FunderInfo memory _funder = funders[sarcoReceiver];
        uint256 usdcCost = _getUsdcCost(
            _funder.sarcoAllocation,
            _funder.usdcToSarcoRateIndex
        );
        return (_funder.sarcoAllocation, usdcCost);
    }

    /**
     * @dev Purchases Sarco for the specified address in exchange for USDC.
     * @notice Sends USDC tokens used to purchase Sarco to Sarco DAO, 
     Approves GeneralTokenVesting contract Sarco Tokens to utilizes allocated Sarco funds,
     Starts token vesting via GeneralTokenVesting contract.
     * @param sarcoReceiver Whitelisted Sarco Investor
     */
    function executePurchase(address sarcoReceiver) external {
        if (offerStartedAt == 0) {
            start();
        }
        require(
            block.timestamp < offerExpiresAt,
            "FundingExecutor: Purchases cannot be made after the offer has expired"
        );

        (uint256 sarcoAllocation, uint256 usdcCost) = getAllocation(msg.sender);

        // Check sender's allocation
        require(
            sarcoAllocation > 0,
            "FundingExecutor: sender does not have a SARCO allocation"
        );

        // Clear sender's allocation
        funders[msg.sender].sarcoAllocation = 0;

        // transfer sender's USDC to this contract
        usdcToken.safeTransferFrom(msg.sender, address(this), usdcCost);

        // Dynamically Build finance app's "message" string
        string memory _executedPurchaseString = string(
            abi.encodePacked(
                "Funding Executed by account: ",
                Strings.toHexString(uint160(msg.sender), 20),
                " for account: ",
                Strings.toHexString(uint160(sarcoReceiver), 20),
                ". Total SARCOs: ",
                Strings.toString(sarcoAllocation),
                "."
            )
        );

        // Forward USDC cost of the purchase to the DAO contract via the Finance Deposit method
        sarcoDao.deposit(address(usdcToken), usdcCost, _executedPurchaseString);

        // Call GeneralTokenVesting startVest method
        GeneralTokenVesting(generalTokenVesting).startVest(
            sarcoReceiver,
            sarcoAllocation,
            vestingEndDelay,
            address(sarcoToken)
        );

        emit FundingExecuted(sarcoReceiver, sarcoAllocation, usdcCost);
    }

    /**
     * @dev If recoverUnsoldTokens > 0 after the offer expired, sarco tokens are send back to Sarco Dao via Finance Contract.
     */
    function recoverUnsoldTokens() external {
        require(
            offerStarted(),
            "FundingExecutor: Purchase offer has not yet started"
        );
        require(
            offerExpired(),
            "FundingExecutor: Purchase offer has not yet expired"
        );

        uint256 unsoldSarcoAmount = sarcoToken.balanceOf(address(this));

        require(
            unsoldSarcoAmount > 0,
            "FundingExecutor: There are no Sarco tokens to recover"
        );

        // Dynamically Build finance app's "message" string
        string memory _recoverTokensString = "Recovered unsold SARCO tokens";

        // Forward recoverable SARCO tokens to the DAO contract via the Finance Deposit method
        sarcoDao.deposit(
            address(sarcoToken),
            unsoldSarcoAmount,
            _recoverTokensString
        );

        // zero out token approvals that this contract has given in its constructor
        usdcToken.approve(address(sarcoDao), 0);
        sarcoToken.approve(address(generalTokenVesting), 0);
        sarcoToken.approve(address(sarcoDao), 0);

        emit TokensRecovered(unsoldSarcoAmount);
    }

    /**
     * @dev Remember that only owner can call so be careful when use on contracts generated from other contracts.
     * @param tokenAddress The token contract address
     * @param tokenAmount Number of tokens to be sent
     * @param recipientAddress The address to send tokens to
     */
    function recoverErc20(
        address tokenAddress,
        uint256 tokenAmount,
        address recipientAddress
    ) public onlyOwner {
        IERC20(tokenAddress).safeTransfer(recipientAddress, tokenAmount);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: contracts/interfaces/GeneralTokenVesting.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.11;

interface GeneralTokenVesting {
    function startVest(
        address beneficiary,
        uint256 tokensToVest,
        uint256 vestDuration,
        address tokenAddress
    ) external;
}

// File: contracts/interfaces/Finance.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.11;

interface Finance {
    function deposit(
        address _token,
        uint256 _value,
        string memory _reference
    ) external payable;
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}